

Les bases de la programmation en CAML

1. Introduction

1.1 Qu'est ce que CAML ?

CAML est un langage de programmation développé par l'INRIA depuis 1985 ; il se range dans la catégorie des langages fonctionnels, mais se prête aussi à la programmation impérative (et orientée objet pour OCAML). Il existe deux implémentations du langage : OCAML est la version la plus avancée et la plus complète ; comme tout langage moderne elle possède une importante bibliothèque logicielle à même de répondre à tous les besoins des programmeurs qui utilisent ce langage. CAML LIGHT est une version plus légère, destinée à l'enseignement et possédant une bibliothèque logicielle très limitée. Bien que cette version ne soit plus maintenue depuis 2002, c'est la version préconisée pour les concours, c'est donc celle que nous utiliserons.

La programmation impérative repose sur la notion de machine abstraite possédant une mémoire et des instructions modifiant les états de celle-ci grâce à des affectations successives. Cela nécessite pour le programmeur de connaître en permanence l'état de la mémoire que le programme modifie, ce qui peut se révéler une tâche complexe.

La programmation fonctionnelle va mettre en avant la définition et l'évaluation de fonctions pour résoudre un problème, en interdisant toute opération d'affectation. Dans ce mode de programmation, la mémoire est donc gérée automatiquement et de manière transparente pour le programmeur.

La programmation orientée objet s'articule autour des données d'un programme. Ceux-ci sont des objets constitués d'attributs les caractérisant et de méthodes les manipulant (ces dernières pouvant être fonctionnelles ou impératives), l'ensemble des attributs et méthodes constituant une classe.

FIGURE 1 – Trois grand paradigmes de la programmation moderne.

Vous trouverez sur la page officielle du langage le manuel complet de CAML LIGHT :

<http://caml.inria.fr/pub/docs/manual-caml-light/>

À part pour des besoins très spécifiques, vous n'aurez besoin que du chapitre intitulé *The core library* dans lequel sont expliquées les instructions clés du langage.

1.2 Comment programmer en CAML LIGHT ?

Ce langage n'étant plus maintenu depuis 2002, les outils développés à cette époque commencent à poser des problèmes de compatibilité avec les matériels récents. Heureusement, des bénévoles proposent des solutions pour vous permettre de continuer à coder en CAML LIGHT.

• *The easy way* (pour Windows, Mac OSX ou Linux)

Jean MOURIC, un collègue de Rennes, maintient à jour un environnement de développement intégré regroupant un éditeur de texte (aux possibilités malheureusement assez limitées) et un interprète de commande pour une utilisation interactive de CAML-LIGHT.

<http://jean.mouric.pagesperso-orange.fr/>

Installez WinCaml si vous êtes sous Windows, MacCaml si vous êtes sous Mac OSX et JavaCaml si vous êtes sous Linux.

• *The hard way* (pour Linux ou Mac OSX)

Si vous êtes sous Linux ou OSX, que vous utilisez déjà un éditeur de code et que l'installation d'un logiciel en ligne de commande ne vous effraie pas, vous pouvez souhaiter installer uniquement CAML LIGHT pour ensuite l'utiliser en interaction avec votre éditeur de code favori.

Sous Linux, utiliser le script écrit par un autre collègue, Judicaël COURANT, que vous trouverez à cette adresse :

```
http://judicael.courant.free.fr/2015/02/20/installationcaml.html
```

Sous OSX, installer les binaires précompilés que l'on trouve à l'adresse suivante (la version Intel) :

```
http://caml.inria.fr/distrib/caml-light-0.80/
```

Seul inconvénient sous OSX, la version installée ne contient pas toutes les bibliothèques et en particulier le module graphique, mais a priori vous n'en aurez pas besoin¹.

Une fois l'installation terminée, ouvrez un terminal et vérifiez que vous pouvez ouvrir une session CAML à l'aide de l'instruction `caml-light`. Il vous reste à lier cette commande à votre éditeur de code (si vous utilisez Emacs, installez le mode Tuareg, qui est un excellent éditeur de code CAML).

• Utilisation interactive

CAML est un langage compilé : on écrit les programmes à l'aide d'un éditeur de textes, et le tout est traité par un compilateur pour créer un exécutable. Il existe cependant une boucle interactive qui permet d'interpréter le langage : l'utilisateur tape des morceaux de programme qui sont traités instantanément par le système, lequel les compile, les exécute et écrit les résultats à la volée. C'est un mode privilégié pour l'apprentissage du langage. Lors d'une session interactive, le caractère `#` qui apparaît à l'écran est le symbole d'invite du système (le *prompt*). Le texte écrit par l'utilisateur commence après ce caractère et se termine par deux points-virgules consécutifs. Une fois lancé, l'interprète va afficher les types et valeurs des expressions évaluées, avant de réafficher le prompt. Commençons donc par observer une première session CAML :

```
> Caml Light version 0.75

# let rec fact = function (* définition d'une fonction *)
  | 0 -> 1
  | n -> n * fact (n-1) ;;
fact : int -> int = <fun>
# fact 5 ;; (* application d'une fonction *)
- : int = 120
# fact 2.3 ;; (* une erreur de typage *)
Toplevel input:
>fact 2.3 ;;
>      ^^^
This expression has type float, but is used with type int.
#
```

Dans ce premier exemple, nous avons commencé par définir une fonction `fact` à l'aide de l'instruction de nommage `let ... = ...`. Cette définition est récursive, comme l'indique le mot-clé `rec`, et procède par filtrage (nous reviendrons plus loin sur cette notion), ce qui nous permet d'obtenir une définition très proche de la définition mathématique de la fonction factorielle :

$$n! = \begin{cases} 1 & \text{si } n = 0 \\ n \times (n-1)! & \text{sinon} \end{cases}$$

L'interprète nous répond que nous avons défini un nouveau nom (`fact`), qui est de type `int -> int`, et que ce nom a pour valeur une fonction (`<fun>`).

Nous pouvons observer que CAML est pourvu d'une reconnaissance automatique de type : cette fonction ne peut s'appliquer qu'à des entiers, et retournera toujours un entier. Nous en avons l'illustration dans la suite de l'exemple : à la commande suivante, l'interprète nous répond que nous avons simplement calculé une valeur sans la nommer (signalé par le caractère `-`), que cette valeur est de type `int`, et qu'elle vaut `120`. En revanche, appliquer cette fonction à un nombre non entier (ici de type `float`) conduit à un message d'erreur. CAML est en effet un langage fortement typé : les erreurs de typage sont détectées et les conversions implicites de type sont formellement interdites.

1. Il est possible de les installer, mais c'est compliqué. Me demander si nécessaire.

1.3 Typage fort, typage faible

Tout objet défini par un langage de programmation possède un *type* (*int*, *float*, etc) qui caractérise la manière dont il sera représenté en mémoire. Le typage du langage CAML est dit *fort* car il possède les particularités suivantes :

- le typage d’une fonction est réalisé au moment de sa définition ;
- les conversions implicites de type sont formellement interdites.

À l’inverse, le typage du langage PYTHON est faible : une même fonction peut s’appliquer à des objets de type différent. Par exemple, la fonction PYTHON `len` calcule indifféremment la longueur d’une chaîne de caractères ou d’une liste alors qu’en CAML existent deux fonctions distinctes `string_length` et `list_length`.

La conversion implicite de type est aussi possible en PYTHON. Par exemple, nous savons que entiers et flottants sont représentés différemment en mémoire, et qu’en conséquence de quoi les algorithmes d’addition entre entiers ou entre flottants ne sont pas les mêmes. Ainsi, lorsqu’en PYTHON vous additionnez un flottant avec un entier, ce dernier est au préalable converti en flottant avant d’être additionné avec l’algorithme d’addition dédié au type *float*². En CAML, cette conversion implicite est impossible, il faut réaliser explicitement la conversion :

```
# 1.0 + 2 ;;
Toplevel input:
> 1.0 + 2 ;;
> ^^^
This expression has type float, but is used with type int.
# 1.0 +. float_of_int 2 ;;
- : float = 3.0
```

On peut observer en outre que CAML n’autorise pas la surcharge d’opérateur : en PYTHON, le symbole `+` désigne différentes fonctions : l’addition des entiers lorsqu’il est situé entre deux entiers, l’addition des flottants lorsqu’il est situé entre deux flottants (ou entre un entier et un flottant), voire la concaténation lorsqu’il est situé entre deux chaînes de caractères. En CAML, l’addition des entiers se note `+`, l’addition des flottants `+.` , la concaténation des chaînes de caractères `^`, etc.

1.4 Définitions globales et locales

En PYTHON, le nommage diffère suivant qu’on nomme un simple objet (avec `=`) ou une fonction (avec `def`). Dans un langage fonctionnel, cette différence s’estompe : *les fonctions sont des objets comme les autres*. En CAML, l’instruction `let` attribue un nom à une valeur. Ces définitions ne sont pas modifiables, elles diffèrent donc fondamentalement du mécanisme d’affectation propre à la programmation de style impératif que nous étudierons plus tard. Une fois défini, un nouveau nom est utilisable dans d’autres calculs :

```
# let n = 12 ;;
n : int = 12
# let f = function x -> x * x ;;
f : int -> int = <fun>
# f n ;;
- : int = 144
```

Les définitions précédentes sont *globales* : la liaison qui a été établie entre le nom et sa valeur est permanente. La syntaxe « `let ... = ... in` » permet de définir temporairement un nom pour la seule durée du calcul en question :

```
# let n = 11 in f n ;;
- : int = 121
# n ;;
- : int = 12
```

On peut observer sur l’exemple précédent que même si elles portent le même nom, variables locale et globale restent bien différenciées.

Notons enfin que le mot `and` permet des définitions multiples :

```
# let n = 9 and f = function x -> x * x * x ;;
n : int = 9
f : int -> int = <fun>
```

2. La réalité est même un peu plus complexe, mais là n’est pas le sujet.

Attention, les valeurs ne deviennent visibles qu'après toutes les déclarations simultanées :

```
# let a = 5 and f = function x -> a * x ;;
Toplevel input:
> let a = 5 and f = function x -> a * x ;;
>
The value identifier a is unbound.
```

Pour réaliser cette définition il faudrait écrire :

```
# let f = let a = 5 in function x -> a * x ;;
f : int -> int = <fun>
```

1.5 Fonctions

Définir une fonction en CAML est simple et proche de la notation mathématique usuelle : on peut utiliser le constructeur **function** (nous y reviendrons plus loin) mais aussi la syntaxe **let f arg = expr**, où **arg** désigne l'argument de la fonction et **expr** le résultat souhaité. Par exemple, pour définir la fonction $f : x \mapsto x + 1$ on écrira :

```
# let f x = x + 1 ;;
f : int -> int = <fun>
# f 2 ;;
- : int = 3
```

On peut noter que l'usage des parenthèses n'est ici pas nécessaire :

$f x$ est équivalent à $f(x)$.

En revanche, elles peuvent se révéler indispensables pour des expressions plus complexes :

```
# f 2 * 3 ;;
- : int = 9
# (f 2) * 3 ;;
- : int = 9
# f (2 * 3) ;;
- : int = 7
```

Cet exemple permet de mettre en évidence le fait que les expressions sont associées à gauche :

$f x y$ est équivalente à $(f x) y$.

• Fonctions à plusieurs arguments

Les fonctions possédant plusieurs arguments se définissent à l'aide de la syntaxe **let f args = expr**, où cette fois-ci **args** désigne les différents arguments de la fonction, séparés par un espace. Par exemple, pour définir l'application $g : (x, y) \mapsto x + y$ on écrira :

```
# let g x y = x + y ;;
g : int -> int -> int = <fun>
# g 2 3 ;;
- : int = 5
```

Attention, il existe ici une ambiguïté : doit-on considérer que g est une fonction à deux variables définies dans \mathbb{Z} , ou que g est une fonction à une variable définie dans $\mathbb{Z} \times \mathbb{Z}$? Observons ce que répond l'interprète de commande à cette deuxième sollicitation :

```
# let h (x, y) = x + y ;;
h : int * int -> int = <fun>
# h (2, 3) ;;
- : int = 5
```

Le type $int * int$ décrit l'ensemble des couples d'entiers, il s'agit donc bien d'une fonction à une variable. Mathématiquement il est facile d'identifier ces deux définitions ; il n'en est pas de même en informatique, et

les langages fonctionnels privilégient la première définition car elle possède un avantage sur la seconde : elle permet aisément de définir des applications partielles. Par exemple, on peut définir la fonction $f : x \mapsto x + 1$ à partir de la fonction g en écrivant : `let f x = g 1 x`, ou encore `let f x = (g 1) x` suivant la règle de l'association à gauche. Le langage CAML autorisant la *simplification à droite*, il nous est alors loisible de définir f de la façon suivante :

```
# let f = g 1 ;;
f : int -> int = <fun>
```

Nous n'aurions pas pu procéder à une telle simplification avec la fonction h , car l'écriture `let f x = h (1, x)` ne permet pas de simplification à droite.

La fonction g est dite *curryfiée*³ par opposition à la fonction h , non curryfiée.

● Fonctions anonymes

Une fonction est donc un objet typé, qui en tant que tel peut être calculé, passé en argument, retourné en résultat. Pour ce faire, on peut construire une valeur fonctionnelle anonyme en suivant la syntaxe `function x -> expr`. Par exemple, les fonctions f , g et h peuvent aussi être définies de la façon suivante :

```
# let f = function x -> x + 1 ;;
f : int -> int = <fun>
# let g = function x -> function y -> x + y ;;
g : int -> int -> int = <fun>
# let h = function (x, y) -> x + y ;;
h : int * int -> int = <fun>
```

Cette nouvelle définition de g a le mérite de mettre en évidence son caractère curryfié, autrement dit la raison pour laquelle `f = g 1` : en effet, `g 1` est équivalent à `function y -> 1 + y`.

Autre conséquence de la curryfication, on peut observer que le typage est associatif à droite :

`int -> int -> int` est équivalent à `int -> (int -> int)`.

Remarque. Par essence, l'instruction `function` ne s'utilise que pour des fonctions d'une variable, ce qui conduit à une écriture un peu lourde pour définir de manière anonyme une fonction curryfiée telle la fonction g . C'est la raison pour laquelle il existe une instruction `fun` qui permet d'alléger l'écriture :

`fun a b c ...` est équivalent à `function a -> function b -> function c -> ...`

Par exemple, la fonction g peut aussi être définie par : `let g = fun x y -> x + y`.

2. Les types en CAML

Nous l'avons déjà dit, toute valeur manipulée par CAML possède un *type* reconnu automatiquement par l'interprète de commande sans qu'il soit nécessaire de le déclarer : connaissant les types des valeurs de base et des opérations primitives, le contrôleur de types produit un type pour une phrase en suivant des règles de typage pour les constructions du langage (nous en avons eu un premier aperçu avec la définition des fonctions). Nous allons maintenant passer en revue les principaux types élémentaires ainsi que quelques opérations primitives, avant de décrire les premières règles de construction des types composés.

2.1 Types élémentaires

● Le vide

Le type *unit* est particulièrement simple, puisqu'il ne comporte qu'une seule valeur, notée `()`, et qu'on appelle *void* (c'est l'équivalent de `None` en `PYTHON`).

Pour comprendre sa raison d'être, imaginons que l'on veuille afficher une chaîne de caractère à l'écran : nous n'avons rien à calculer, seulement modifier l'environnement (c'est à dire réaliser ce qu'on appelle un *effet de bord*). Or un langage fonctionnel ne peut que définir et appliquer des fonctions. Voilà pourquoi la fonction `print_string` est de type `string -> unit` : elle renvoie le résultat *void*, tout en réalisant au passage un effet de bord.

3. Aucun rapport avec la cuisine, cette définition est un hommage au mathématicien Haskell CURRY.

```
# print_string "Hello World" ;;
Hello World- : unit = ()
```

Autre exemple, la fonction `print_newline` est de type `unit -> unit` et a pour effet de passer à la ligne. Pour l'utiliser, il ne faut donc pas oublier son argument (qui ne peut qu'être le `void`) :

```
# print_newline () ;;
- : unit = ()
```

• Les entiers

Sur les ordinateurs équipés d'un processeur 64 bits, les éléments de type `int` sont les entiers de l'intervalle $[-2^{62}, 2^{62} - 1]$, les calculs s'effectuant modulo 2^{63} suivant la technique du complément à deux (voir votre cours d'informatique de tronc commun). Il faut donc veiller à ne pas dépasser ces limites sous peine d'obtenir des résultats surprenants :

```
# let a = 2147483648 in a * a ;;
- : int = -4611686018427387904
# let a = 2147483648 in a * a - 1 ;;
- : int = 4611686018427387903
```

(Vous avez peut-être deviné que $2^{31} = 2\,147\,483\,648$.)

Notez que `max_int` et `min_int` vous donnent les valeurs maximales et minimales de type entier :

```
# max_int ;;
- : int = 4611686018427387903
# min_int ;;
- : int = -4611686018427387904
```

Les opérations usuelles se notent `+`, `-`, `*` (multiplication), `/` (quotient de la division euclidienne) et `mod` (reste de la division euclidienne). Multiplication, quotient et reste ont priorité face à l'addition et la soustraction, dans les autres cas la règle d'association à gauche s'applique, et on dispose des parenthèses pour hiérarchiser les calculs si besoin est :

```
# 1 + 3 mod 2 ;;      (* 1 + (3 mod 2) *)
- : int = 2
# 2 * 5 / 3 ;;      (* (2 * 5) / 3 *)
- : int = 3
# 5 / 3 * 2 ;;      (* (5 / 3) * 2 *)
- : int = 2
```

• Les réels

Les nombres réels sont approximativement représentés par le type `float`. Tout comme en `PYTHON`, on les définit sous forme décimale (par exemple `3.141592653`) éventuellement accompagnés d'un exposant représentant une puissance de 10 (par exemple `1.789e3`). Les calculs effectués sur les flottants sont bien évidemment approchés. Les opérateurs usuels se notent `+. , -. , *. , /.` (noter le point qui permet de les différencier des opérateurs sur le type `int`), et `**` (élévation à la puissance). On dispose en outre d'un certain nombre de fonctions mathématiques telles : `sqrt` (racine carrée), `log` (qui désigne le logarithme népérien), `exp`, `sin`, `cos`, `tan`, `asin` (arcsinus), `acos`, `atan`, etc.

```
# let a = 3.141592654 in tan (a /. 4.) ;;
- : float = 1.00000000021
```

Attention à ne pas oublier que `CAML` est un langage fortement typé : il n'y a pas de conversion implicite de type, et vous ne pouvez donc pas appliquer une fonction du type `float` à une valeur de type `int` :

```
# let a = 3.141592653 in tan (a /. 4) ;;
Toplevel input:
>let a = 3.141592653 in tan (a /. 4) ;;
>
This expression has type int, but is used with type float.
```

Le calcul numérique relevant du programme d'informatique de tronc commun nous n'aurons que rarement besoin de manipuler des flottants.

• Caractères et chaînes de caractères

Les caractères sont les éléments du type *char*; on les note en entourant le caractère par le symbole ` (backcote, ou accent grave). Les éléments de type *string* sont des chaînes de caractères entourées du symbole " (double quote). Nous reviendrons sur la notion de chaîne de caractère lorsque nous aborderons les vecteurs; pour l'instant, contentons nous de noter quelques fonctions qui pourront nous être utiles :

La concaténation des chaînes de caractères est notée par le symbole ^ :

```
# "liberté, " ^ "égalité, " ^ "fraternité" ;;
- : string = "liberté, égalité, fraternité"
```

La fonction prédéfinie *string_length* renvoie la longueur d'une chaîne de caractères; c'est donc une fonction de type *string* → *int*.

La fonction *nth_char* renvoie le *n*^e caractère d'une chaîne de caractères (attention, dans une chaîne les indices commencent à 0). C'est une fonction de type *string* → *int* → *char*. Notons que si une chaîne porte un nom *s* alors *s.[k]* retourne le caractère d'indice *k* de *s* (*s.[k]* est donc équivalent à *nth_char s k*).

La fonction *sub_string* retourne une sous-chaîne partant d'un indice donné et d'une longueur donnée; c'est une fonction de type *string* → *int* → *int* → *string*.

```
# nth_char "Caml" 1 ;;
- : char = 'a'
# sub_string "informatique" 2 6 ;;
- : string = "format"
```

• Les booléens

Le type *bool* comporte deux valeurs : le vrai (*true*) et le faux (*false*), et dispose des opérations logiques non (*not*) et (&&) ou (||). Les deux derniers opérateurs fonctionnent suivant le principe de l'évaluation paresseuse : *expr1 && expr2* ne va évaluer la deuxième expression que si la première est vraie, et *expr1 || expr2* ne va évaluer la deuxième expression que si la première est fautive.

```
# not (1 = 2) || (1 / 0 = 1) ;;
- : bool = true
# not (1 = 2) && (1 / 0 = 1) ;;
Uncaught exception: Division_by_zero
```

• Les paires

Il est possible de définir le produit cartésien de deux types : si *x* est un élément de type '*a*' et *y* un élément de '*b*', alors (*x*, *y*) est un élément de type '*a***b*'.

Par exemple, nous avons vu que la version non curriifiée de la fonction (*x*, *y*) ↦ *x* + *y* se définit par :

```
# let h (x, y) = x + y ;;
h : int * int -> int = <fun>
```

Il est bien entendu possible de faire le produit cartésien d'un nombre plus important de types, mais seuls les paires disposent des fonctions *fst* et *snd* pour récupérer la première et la seconde composante du couple :

```
# fst ("caml", 0.75) ;;
- : string = "caml"
# snd ("caml", 0.75) ;;
- : float = 0.75
```

2.2 Polymorphisme

On a pu constater que CAML est pourvu d'une reconnaissance automatique de type : par exemple, c'est la présence de l'entier `1` et de l'opérateur entier `+` qui permet d'associer à la fonction `function n -> n + 1` le type `int -> int`.

Mais il peut arriver qu'une fonction puisse s'appliquer indifféremment à tous les types ; on dit dans ce cas que cette fonction est *polymorphe*. On utilise alors les symboles `'a`, `'b`, `'c`,... pour désigner des types quelconques. Revenons par exemple sur les fonctions `fst` et `snd` qui agissent sur les paires. Leur définition est évidente, et n'imposent aucune contrainte de typage :

```
# let fst (x, y) = x ;;
fst : 'a * 'b -> 'a = <fun >
# let snd (x, y) = y ;;
snd : 'a * 'b -> 'b = <fun >
```

Nous avons vu que les opérations algébriques, l'addition par exemple, ne sont pas polymorphes : l'opérateur infixe `+` est de type `int -> int` alors que `+.` est de type `float -> float`. En revanche, les opérations de comparaison telle l'égalité (notée `=`) le sont⁴. On peut utiliser le même opérateur pour comparer entre eux des entiers, des nombres complexes ou des chaînes de caractères (et plus généralement tout objet pour lesquels la comparaison a un sens), car c'est leur représentation machine qui est comparée.

```
# 1 = 1 ;;
- : bool = true
# 1.0 = 2.0 ;;
- : bool = false
# prefix = ;;
- : 'a -> 'a -> bool = <fun >
```

Les autres opérateurs de comparaison polymorphes sont `<` (la négation de l'égalité), `<`, `>`, `<=` (inférieur ou égal), `>=` (supérieur ou égal).

```
# "anaconda" < "zebre" ;;
- : bool = true
```

(La comparaison des chaînes de caractères utilise l'ordre lexicographique.)

2.3 Types construits

À partir de types existant on peut construire de nouveaux types à l'aide de la syntaxe : `type montype = constr` où `constr` est une construction de type utilisant :

- des noms (analogues à des constantes) ;
- des noms avec paramètre en suivant la syntaxe : `nom of type` ;
- l'opérateur `|` correspondant au « ou » logique (ce qu'on appelle un type somme) ;
- des types produits (analogue au « et » logique) obtenus en dressant la liste des caractéristiques suivant la syntaxe : `{ car1 : type1 ; ... ; carN : typeN }`.

Par exemple, s'il n'existait déjà, le type `bool` serait défini par : `type bool = true | false`.

On peut donner un exemple un peu plus élaboré de type somme en définissant l'analogue de la droite numérique achevée $\overline{\mathbb{R}} = \mathbb{R} \cup \{-\infty, +\infty\}$:

```
# type reel_etendu = Reel of float | Plus_infini | Moins_infini ;;
Type reel_etendu defined.
# Plus_infini ;;
- : reel_etendu = Plus_infini
# Reel 3.14 ;;
- : reel_etendu = Reel 3.14
```

et définir une fonction pour plonger \mathbb{R} dans $\overline{\mathbb{R}}$:

4. Vous risquez malheureusement de souvent confondre l'égalité en CAML qui se note `=` et l'égalité en PYTHON qui se note `==`. Et c'est la même chose avec l'opérateur de différence (`<>` en CAML et `!=` en PYTHON).

```
# let etendu_of_float x = Reel x ;;
etendu_of_float : float -> reel_etendu = <fun>
# etendu_of_float 3.14 ;;
- : reel_etendu = Reel 3.14
```

La définition d'un type associé aux nombres complexes est un bon exemple de type produit :

```
# type complexe = {Re : float ; Im : float} ;;
Type complexe defined.
```

Pour construire une valeur de ce type, on énumère ses caractéristiques particulières :

```
# let i = {Re = 0. ; Im = 1.} ;;
i : complexe = {Re=0.0; Im=1.0}
# let complexe_of_float x = {Re = x ; Im = 0.} ;;
complexe_of_float : float -> complexe = <fun>
```

Pour accéder à l'une des caractéristiques d'un objet de type produit, il suffit de faire suivre le nom d'un point et du nom de la caractéristique. Définissons par exemple la multiplication des nombres complexes :

```
# let mult x y =
  let a = x.Re and b = x.Im and c = y.Re and d = y.Im in
  {Re = a*.c-.b*.d ; Im = a*.d+.b*.c} ;;
mult : complexe -> complexe -> complexe = <fun>
# mult i i ;;
- : complexe = {Re=-1.0; Im=0.0}
```

• Types récurifs

Il est possible de définir un type récursif, c'est-à-dire intervenant dans sa propre définition. Nous allons à titre d'exemple définir les couleurs par synthèse soustractive, en convenant qu'une couleur est une couleur primaire ou alors un mélange de deux couleurs :

```
# type couleur = Cyan | Magenta | Jaune | Melange of couleur * couleur ;;
Type couleur defined.
```

Cette définition récursive nous permet de définir de nouvelles couleurs :

```
# let rouge = Melange (Magenta, Jaune) ;;
rouge : couleur = Melange (Magenta, Jaune)
# let orange = Melange (Rouge, Jaune) ;;
orange : couleur = Melange (Melange (Magenta, Jaune), Jaune)
```

Écrivons maintenant une fonction qui calcule la composante CMJ d'une couleur :

```
# let rec cmj = function
  | Cyan          -> (1., 0., 0.)
  | Magenta       -> (0., 1., 0.)
  | Jaune         -> (0., 0., 1.)
  | Melange (coul1, coul2) -> let (a, b, c) = cmj coul1 and (d, e, f) = cmj coul2
                               in ((a+.d)/.2., (b+.e)/.2., (c+.f)/.2.) ;;
cmj : couleur -> float * float * float = <fun>
# cmj orange ;;
- : float * float * float = 0.0, 0.25, 0.75
```

3. Définition d'une fonction

Nous l'avons déjà dit, une fonction se définit à l'aide de la syntaxe `let f = function ... -> ...` et son type est alors de la forme `type1 -> type2`, le typage étant implicite, et polymorphe le cas échéant. Cette syntaxe est en outre équivalente à `let f ... = ...`.

Rappelons aussi que `let f = fun x y -> ...` est la forme raccourcie de :

```
let f = function x -> function y -> ...
```

et qu'on peut aussi écrire `let f x y = ...`. Son type est alors *type_de_x -> type_de_y -> type_du_résultat*.

3.1 Analyse par cas : le filtrage

Le filtrage est un trait extrêmement puissant de CAML, et à ce titre très fréquemment employé. La syntaxe d'une fonction travaillant par filtrage peut prendre une des formes suivantes :

```
let f = function
| motif1 -> expr1
| motif2 -> expr2
| .....
| motifn -> exprn ;;
```

```
let f x = match x with
| motif1 -> expr1
| motif2 -> expr2
| .....
| motifn -> exprn ;;
```

On parle de filtrage *implicite* dans le premier cas et de filtrage *explicite* dans le second.

Lors du calcul d'une telle fonction, l'interprète de commande va essayer de faire concorder l'argument avec les motifs successifs qui apparaissent dans la définition. C'est l'expression correspondant au premier motif reconnu qui sera calculée. Bien entendu, il est nécessaire que les différentes expressions du résultat soient du même type.

Lorsque le motif est un nom, il s'accorde avec n'importe quelle valeur, et la reçoit dans l'expression exécutée. Par exemple, la définition de la fonction sinus cardinal sera :

```
# let sinc = function
| 0. -> 1.
| x -> sin (x) /. x ;;
sinc : float -> float = <fun>
```

Le motif `0.` est une constante qui ne concorde qu'avec la valeur `0.` ; le motif `x` est une variable qui concorde avec n'importe quel nombre flottant.

Lorsqu'il n'est pas nécessaire de nommer un motif (ou lorsqu'il a déjà été nommé précédemment, par exemple lors d'un filtrage explicite), on utilise le caractère `_` qui s'accorde avec n'importe quelle valeur. Par exemple, la fonction `sinc` peut aussi s'écrire :

```
# let sinc x = match x with
| 0. -> 1.
| _ -> sin (x) /. x ;;
sinc : float -> float = <fun>
```

Attention, l'ordre dans lequel on essaye de faire correspondre un motif et une valeur a de l'importance. Par exemple, la définition ci-dessous est incorrecte, car la valeur `0.` s'accorde avec le premier motif :

```
# let sinc = function
| x -> sin(x) /. x
| 0. -> 1. ;;
Toplevel input:
> | 0. -> 1. ;;
>   ^^
Warning: this matching case is unused.
sinc : float -> float = <fun>

# sinc 0. ;;
- : float = nan.0
```

(Notez que l'interprète de commande indique que le deuxième motif du filtrage est inutile).

Attention à ne pas non plus confondre concordance avec un motif et égalité avec un nom prédéfini. Considérons la fonction suivante, qui détermine si un entier est nul :

```
# let est_nul = function
| 0 -> true
| _ -> false ;;
est_nul : int -> bool = <fun>
```

Cette définition est parfaitement correcte, mais essayons de la généraliser en écrivant :

```
# let egal x = function
  | x -> true
  | _ -> false ;;
Toplevel input:
> | _ -> false ;;
>   ^
Warning: this matching case is unused.
egal : 'a -> 'b -> bool = <fun>
```

Deux indices nous apprennent que cette fonction ne répondra vraisemblablement pas à notre attente : son type, qui devrait être `'a -> 'a -> bool`, et le message de l'interprète indiquant que le second motif ne sera jamais utilisé. En effet, il ne faudra jamais perdre de vue le fait que le filtrage est structurel : on ne peut utiliser dans les motifs que des constructeurs, des constantes et des noms, à l'exclusion des valeurs calculées.

En tout état de cause, tout se passe comme si nous avions écrit :

```
let egal x = function y -> true ;;
```

Le filtrage par motif (*pattern matching*) permet de vérifier si l'objet du filtrage possède une structure donnée ; ce n'est pas un test d'égalité avec un nom prédéfini.

• Motif gardé

Pour corriger notre fonction, il est possible d'utiliser un *motif gardé* dont la syntaxe est :

```
| motif when condition ->
```

Dans un tel cas de filtrage, le motif n'est reconnu que si la condition (une expression de type `bool`) est vérifiée. Nous pouvons donc corriger la fonction précédente en :

```
# let egal x = function
  | y when x = y -> true
  | _ -> false ;;
egal : 'a -> 'a -> bool = <fun>
```

Remarque. Cet exemple a pour but de vous faire comprendre la différence entre test d'égalité et concordance avec un motif. Il n'en reste pas moins que la meilleure définition de cette fonction reste sans doute :

```
let egal x y = x = y ;;
```

ou plus simplement encore :

```
let egal = prefix = ;;
```

(l'instruction `prefix` transforme un opérateur infix en opérateur préfixe.)

Nous venons de voir que l'interprète de commande est capable de repérer des motifs de filtrage superflus. Il est aussi capable de repérer des filtrages incomplets. Observons par exemple cette première tentative pour définir l'addition sur les réels étendus (définis précédemment) :

```
# let addition r s = match (r, s) with
  | (Reel x, Reel y) -> Reel (x +. y)
  | (Moins_infini, Reel x) -> Moins_infini
  | (Plus_infini, Reel x) -> Plus_infini
  | (Reel x, Moins_infini) -> Moins_infini
  | (Reel x, Plus_infini) -> Plus_infini ;;
Warning: this matching is not exhaustive.
addition : reel_etendu -> reel_etendu -> reel_etendu = <fun>
```

En effet, nous n'avons pas encore défini l'addition des infinis entre eux :

```
# addition Plus_infini Plus_infini ;;
Uncaught exception: Match_failure ("", 1131, 1373)
```

il nous faut ajouter les règles $(+\infty) + (+\infty) = (+\infty)$ et $(-\infty) + (-\infty) = -\infty$, et lever une exception dans le cas de l'addition entre $+\infty$ et $-\infty$. Faisons une nouvelle tentative :

```
# let addition r s = match (r, s) with
  | (Reel x, Reel y)      -> Reel (x +. y)
  | (Moins_infini, Reel x) -> Moins_infini
  | (Plus_infini, Reel x)  -> Plus_infini
  | (Reel x, Moins_infini) -> Moins_infini
  | (Reel x, Plus_infini)  -> Plus_infini
  | (a, a)                 -> a
  | _                       -> failwith "opération non définie" ;;
Toplevel input:
> | a a                    -> a
> | ^
The variable a is bound several times in this pattern.
```

Nous observons ici une autre contrainte du filtrage : il n'est pas possible d'utiliser dans un motif le même nom à plusieurs reprises. Là encore, l'utilisation d'un motif gardé résout le problème :

```
# let addition r s = match (r, s) with
  | (Reel x, Reel y)      -> Reel (x +. y)
  | (Moins_infini, Reel x) -> Moins_infini
  | (Plus_infini, Reel x)  -> Plus_infini
  | (Reel x, Moins_infini) -> Moins_infini
  | (Reel x, Plus_infini)  -> Plus_infini
  | (a, b) when a = b      -> a
  | _                       -> failwith "opération non définie" ;;
addition : reel_etendu -> reel_etendu -> reel_etendu = <fun>
```

Attention, la vérification de l'exhaustivité d'un filtrage dans lequel un motif est gardé suppose que l'expression conditionnelle attachée à celui-ci est fautive; en conséquence de quoi, ce motif ne sera pas pris en compte. Par exemple, CAML ne peut détecter que le filtrage suivant est exhaustif :

```
# let f = function
  | x when x >= 0 -> x + 1
  | x when x < 0 -> x - 1 ;;
Toplevel input:
Warning: this matching is not exhaustive.
f : int -> int = <fun>
```

En règle générale, laisser des filtrages non exhaustifs dans un programme est considéré comme un laisser-aller de mauvais aloi. Dans la mesure du possible, il est préférable que le dernier motif d'un filtrage soit un motif générique non gardé de manière à assurer l'exhaustivité du filtrage. dans le cas ci-dessus on se contentera donc d'écrire :

```
# let f = function
  | x when x >= 0 -> x + 1
  | x              -> x - 1 ;;
f : int -> int = <fun>
```

3.2 Récursivité

Considérons de nouveau la définition de la fonction factorielle que nous avons donnée en introduction :

```
let rec fact = function
  | 0 -> 1
  | n -> n * fact (n-1) ;;
```

Le mot clé `rec` indique que nous avons défini un objet *récurif* (ici une fonction), c'est-à-dire un objet dont le nom intervient dans sa définition. Nous approfondirons plus tard cette notion, très importante dans un langage fonctionnel (la majorité des fonctions que nous écrirons seront récursives). Dans l'immédiat, contentons-nous d'observer sa proximité avec nombre de définitions mathématiques, à commencer avec les suites récurrentes.

Notons enfin qu'il est possible de définir deux fonctions mutuellement récursives à l'aide de la syntaxe `let rec f = ... and g = ...`.

Les deux fonctions suivantes, par exemple, déterminent la parité d'un entier naturel (de manière assez inefficace il faut bien dire) :

```
# let rec est_pair = function
  | 0 -> true
  | n -> est_impair (n-1)
and est_impair = function
  | 0 -> false
  | n -> est_pair (n-1) ;;
est_pair : int -> bool = <fun>
est_impair : int -> bool = <fun>
```

4. Exercices

4.1 Définitions globales et locales

Exercice 1 Prévoir la réponse de l'interprète de commande après les définitions suivantes :

```
# let a = 2 ;;
a : int = 2
# let f x = a * x ;;
f : int -> int = <fun>
# let a = 3 in f 1 ;;
???
```

```
# let a = 3 and f x = a * x ;;
a : int = 3
f : int -> int = <fun>
# f 1 ;;
???
```

Exercice 2 Prévoir la réponse de l'interprète de commande après les deux définitions suivantes :

```
# let a =
  let a = 3 and b = 2 in
  let a = a + b and b = a - b in
  a - b ;;
???
```

```
# let b = 2 in a - b * b ;;
???
```

4.2 Typage

Exercice 3 Donner des expressions CAML qui correspondent aux types suivants :

- $(int \rightarrow int) \rightarrow int$
- $int \rightarrow (int \rightarrow int)$
- $int \rightarrow int \rightarrow int$
- $int \rightarrow (int \rightarrow int) \rightarrow int$

Exercice 4 Déterminer sans utiliser l'interprète de commandes le type des expressions suivantes :

- `fun f x y -> f x y;`
- `fun f g x -> g (f x);`
- `fun f g x -> (f x) + (g x).`

Exercice 5 Deviner la réponse de l'interprète de commande lorsqu'on saisit l'expression :

```
# let h (f, g) = function x -> f (g x) ;;
???
```

Exercice 6 Les exercices précédents ont été facilités par le choix des lettres **f** et **g** pour désigner implicitement des fonctions et **x** et **y** pour des variables. Oublions maintenant cette aide pour déterminer le type des expressions suivantes :

- a) `fun x y z -> (x y) z;`
- b) `fun x y z -> x (y z);`
- c) `fun x y z -> x y z;`
- d) `fun x y z -> x (y z x);`
- e) `fun x y z -> (x y) (z x).`

4.3 Définition de fonctions

Exercice 7 Définir une fonction de type `bool -> bool -> bool` représentant l'opérateur logique \Rightarrow .

Exercice 8 L'opérateur des *différences finies* Δ associe à toute suite $(u_n)_{n \in \mathbb{N}}$ la suite $(u_{n+1} - u_n)_{n \in \mathbb{N}}$. Écrire une fonction `delta` qui réalise cette transformation. On commencera par en donner son type.

Exercice 9 Définir une fonction `curry` qui transforme une fonction à deux variables non curryfiée en une fonction curryfiée, puis une fonction `uncurry` qui réalise la transformation inverse.

Exercice 10 Un nombre de HAMMING est un entier qui ne comporte que des 2, 3 et 5 dans sa décomposition en facteurs premiers. Écrire une fonction `hamming` de type `int -> bool` qui détermine si un entier donné est un nombre de HAMMING.

Exercice 11 Définir en CAML les deux fonctions suivantes :

$$f : (p, q) \mapsto \begin{cases} q & \text{si } p = 0 \\ f(p-1, q) + 1 & \text{sinon} \end{cases} \quad \text{et} \quad g : (p, q) \mapsto \begin{cases} q & \text{si } p = 0 \\ g(p-1, q+1) & \text{sinon} \end{cases}$$

Que calculent-elles ?

Exercice 12 Définir en CAML une fonction récursive calculant le produit de deux entiers naturels en suivant le principe de la multiplication égyptienne, qui utilise les formules :

$$p \times q = \begin{cases} \left(\frac{p}{2}\right) \times (2q) & \text{si } p \text{ est pair} \\ \left(\frac{p-1}{2}\right) \times (2q) + q & \text{si } p \text{ est impair} \end{cases}$$

Rédiger en suivant le même principe une fonction qui calcule q^p lorsque p et q sont des entiers.